
Abstract 

Organizations around the world develop business intelligence and analytics dashboards, sometimes referred to as enterprise 
dashboards, to display the status of “point-in-time” metrics and key performance indicators. Effectively designed dashboards 
extract real-time data from multiple sources for the purpose of highlighting important information, numbers, tables, statistics, 
metrics, performance scorecards and other essential content. This paper explores essential rules for “good” dashboard design, 
the metrics frequently used in dashboards, and the use of best practice programming techniques in the design of quick and 
easy dashboards using SAS® software. Learn essential programming techniques to create real-world dashboards using Base-
SAS® software including PROC SQL, macro, Output Delivery System (ODS), ODS HTML, ODS Excel, ODS Layout, ODS Statistical 
Graphics, PROC SGPLOT, PROC SGPIE, and other technologies. 
 

Introduction 

In a world of big data where data repositories and the demand placed on them are growing at explosive levels, organizations 
are faced with a number of decisions related to their information requirements: 
 

1) What are the best ways to handle large amounts of information? 

2) How should analytical data be processed? 

3) What are the choices for constructing the most effective information delivery mechanisms? 

4) How should analytical data and results be displayed? 

 
To help answer these and other questions, this paper explains what a dashboard is, the dashboard’s elements, the do’s and 
don’ts for constructing effective dashboards, dashboard design techniques, an investigation of the various types of dashboards, 
the merits and strengths of using the base-SAS® software to construct dashboards, and an illustration of a few dashboard 
examples along with the base-SAS code used in their construction. 
 

Example Table 

The dashboard examples displayed in this paper reference the dataset (or table), SASHELP.CARS. The SASHELP.CARS dataset 

consists of 428 observations and 15 variables and is illustrated below. 

 

SASHELP.CARS Table 

“Brief” History of Dashboards 

In the world of information technology, a dashboard serves as a user interface to organize and display information visually in 
the simplest way possible. Dashboards originated in the 1970’s as decision support tools and systems that served management, 
operations, and organizational planning.  In the 1980’s, dashboards came of age as executive information systems emphasizing 
graphical displays and simple user interfaces to assist with management decision making. In the 1990’s, dashboards 
experienced a growing interest with the rise of the Internet. As information technology and the Internet entered the 2000’s, 
vendors including SAS Institute, and others, offered high-end easy-to-use products for the development of comprehensive 
“custom” dashboards. The dashboards being built today offer users the ability to monitor key metrics, information summaries, 
and reports in a single easy-to-use user interface. As a result, dashboards are designed to alert users to key business issues that 
impact an organization’s tactics and strategies by facilitating improved decision making activities. 
 
So exactly what is a dashboard? In the paper, “Building Your First Dashboard Using the SAS® 9 Business Intelligence Platform: A 
Tutorial,” by Gregory S. Nelson (2009), Nelson describes a dashboard as a visualization technique that provides an immediate 
view or snapshot of exactly where you are in a specific process relative to your stated goals and objectives. He adds that, Visual 
indicators, such as temperature gauges, traffic lights and speedometers, help give a real-world sense of present progress and 
assists in making decisions, adapting to current conditions or drilling into more detailed information. As a user interface, 
dashboards display performance indicators (PIs), key performance indicators (KPIs), and other relevant information. 
 

Types of Dashboards 

The first step in dashboard design is to understand the purpose and type of dashboard you will need. With three types of 
dashboard designs available, users are encouraged to select the dashboard type that best meets your needs. The following 
table describes the three types of enterprise dashboards and their purpose. 
 

Dashboard Type Purpose 

Strategic Dashboards 

Strategic dashboards provide executives and managers with visual information to determine and 
support goals and objectives within an organization. This type of dashboard facilitates monitoring 
an organization’s health, progress, performance, and areas where improvement can be made. 
There is typically no need for interactive features with this type of dashboard. Strategic 
dashboard examples include: Sales, Human Resources, Manufacturing, and Services. 

Analytical Dashboards 

Analytical dashboards provide users with visual information to help gain a better understanding 
with historical, present and future data; understand trends; allow comparisons to be made; and 
determine the type of adjustments that are needed. Analytical dashboards should allow 
interactive features such as drill-down capabilities, as needed, to access more detailed 
information. Dashboard examples include: obtaining real-time data and information, determining 
why some things are working and others are not, identifying patterns and opportunities with 
your data, and aligning strategic objectives with performance initiatives. 

Operational Dashboards 

Operational dashboards provide users with visual information to concentrate on performance 
monitoring and measurements, monitor the efficiency and effectiveness of their organization. 
There is typically a need to update information displayed in an operational dashboard frequently 
to make it relevant to the users’ needs. Dashboard examples include: improved understanding of 
performance, better focus and alignment, and faster and better decision making. 

 

Dashboard Elements 

In Malik Shadan’s (2007) paper, Elements for an Enterprise Dashboard, he mentions that there are basic and advanced 
characteristics specific to an enterprise dashboard. The basic characteristics encompass the acronym, SMART, and the 
advanced characteristics of an enterprise dashboard encompass the acronym, IMPACT. The elements associated with each 
acronym appear in the following tables. 
 

SMART Basic Elements 

Element Description 

Synergetic Synergize information in a single screen view.  

Monitor KPIs Display critical KPIs for effective decision making. 

Accurate Dashboard must be well tested and validated, and information must be accurate. 

Responsive Respond to user alerts and visual content to draw immediate attention to critical matters. 

Timely Display information that is real-time and right-time for effective decision making. 

IMPACT Advanced Elements 

Element Description 

Interactive Allow user to drill-down and derive details, root causes and more.  

More Data History Allow users to review historical trends for any KPI. 

Personalized Display should be specific to each user’s domain of responsibility, data restrictions, and 
privileges. 

Analytical Allow users to perform guided analysis, compare, contrast, and make analytical inferences. 

Collaborative Facilitate users’ ability to exchange notes regarding observations on their dashboard. 

Trackability Allow each user to customize the metrics they would like to track. 

 

13 Common Pitfalls to Avoid when Designing Dashboards 

Successful dashboard design involves the transformation of quantitative data into meaningful and effective visual displays 
including graphs, maps, gauges and summary information. In his paper, “Common Pitfalls in Dashboard Design,” Stephen Few 
(2006) proposes 13 common mistakes many make when designing dashboards. Instead of concentrating on what should be 
done when designing dashboards, Mr. Few’s body of work espouses the most common mistakes along with detailed 
explanations to help educate current and future designers alike. I have listed the 13 common pitfalls from Mr. Few’s seminal 
work, below, but readers are encouraged to read his entire paper, see the References section, for a complete perspective. 
 
Stephen Few’s 13 Common Pitfalls in Dashboard Design (cited from reference) 
 

Pitfall Description 

Pitfall #1 Exceeding the Boundaries of a Single Screen 

Pitfall #2 Supplying Inadequate Context for the Data 

Pitfall #3 Displaying Excessive Detail or Precision 

Pitfall #4 Expressing Measures Indirectly 

Pitfall #5 Choosing Inappropriate Media of Display 

Pitfall #6 Introducing Meaningless Variety 

Pitfall #7 Using Poorly Designed Display Media 

Pitfall #8 Encoding Quantitative Data Inaccurately 

Pitfall #9 Arranging the Data Poorly 

Pitfall #10 Ineffectively Highlighting What’s Important 

Pitfall #11 Cluttering the Screen with Useless Decoration 

Pitfall #12 Misusing or Overusing Color 

Pitfall #13 Designing an Unappealing Visual Display 

 

Steps to Creating a Dashboard using Base-SAS® Software 

Follow these basic steps to successfully construct a quick and easy dashboard using the SAS software. 
 

1. Connect to desired data sources using Libname statement. 

2. Specify an ODS HTML5 statement to produce dashboards that can be viewed with a web browser. 

3. Specify an ODS LAYOUT statement to tell SAS how many row(s) and column(s) the dashboard should contain. 

4. Specify an ODS REGION statement to control where output is to be placed on the dashboard. 

5. Specify color settings, fonts, font attributes, and other “customizations”. 

6. Specify an ODS LAYOUT END statement to terminate the dashboard layout. 

7. Specify an ODS HTML5 CLOSE statement to render the results to the dashboard file. 

 
Several quick and easy dashboard examples are illustrated below. 

Examples 

 

Example #1 – (1x1) Dashboard Layout with Default Settings 
PROC FREQ, PROC SGPLOT and PROC MEANS 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The two procedures that are 
used to create the dashboard are: PROC FREQ and PROC SGPLOT. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

3. An ODS LAYOUT GRIDDED ROWS=1 COLUMNS=1 statement tells SAS to create a gridded layout consisting of one row 
and ONE column. 
 

4. An ODS REGION statement tells SAS to produce the results using PROC FREQ and PROC SGPLOT. 
 

5. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

6. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 
Base-SAS Code: 
ods html5 path="/home/kirklafler/Dashboards/Results" 

          body="Dashboard - Gridded HTML (1 x 1) Layout.html" 

          (url=none) ; 

 

title1 font=impact bold h=12 c=blue "Analytics Dashboard" ; 

 

ODS LAYOUT GRIDDED ROWS=1 COLUMNS=1 ; /* Design HTML 1x1 Layout */ 

 

options center ; /* Center the Results */ 

ods region ;  /* Start of Output Results */ 

title1 "SASHELP.CARS Frequency Distribution for Origin and Type" ; 

proc freq data=SASHELP.CARS NLEVELS ; 

  table Origin Type ; 

run ; 

 

title1 "Origin BarChart" ; 

proc sgplot data=SASHELP.CARS ; 

  vbar Origin / group=Origin datalabel ; 

run ; 

 

title1 "Type BarChart" ; 

proc sgplot data=SASHELP.CARS ; 

  vbar Type / group=Type datalabel ; 

run ; 

 

title1 "Origin by Type Cluster BarChart" ; 

proc sgplot data=SASHELP.CARS ; 

  vbar Origin / group=Type response=MSRP stat=mean groupdisplay=cluster datalabel ; 

run ; 

 

title1 "Descriptive Statistics for MSRP and Invoice by Origin" ; 

proc means data=SASHELP.CARS n nmiss min max range mean median mode std var ; 

  class Origin Type ; 

run ; 

title ; 

 

ods layout end ; /* Terminate the Layout of Output Results */ 

ods html5 close ; 

Example #2 – (1x2) Dashboard Layout with Default Settings 
PROC FREQ and PROC REPORT 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The two procedures that are 
used to create the dashboard are: PROC FREQ and PROC REPORT. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

3. An ODS LAYOUT GRIDDED ROWS=1 COLUMNS=2 statement tells SAS to create a gridded layout consisting of one row 
and two columns. 
 

4. The first ODS REGION statement tells SAS to produce the first column of results using PROC FREQ. 
 

5. The second ODS REGION statement tells SAS to produce the second column of results using PROC REPORT. 
 

6. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

7. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 
Base-SAS Code: 
ODS HTML5 PATH="/home/kirklafler/Results" 
          FILE="Dashboard #1 - (1x2) Layout.html" 
          (URL=NONE) ; 
 
ODS LAYOUT GRIDDED ROWS=1 COLUMNS=2 ; 
 
PROC SQL NOPRINT ; 
  SELECT NAME 
   INTO :mAlphabeticalVariable_List SEPARATED BY " " 
    FROM SASHELP.VCOLUMN 
     WHERE LIBNAME="SASHELP" AND MEMNAME="CARS"  
      ORDER BY NAME ; 
QUIT ; 
 
ODS REGION ;  /* Row 1 Column 1 */ 
ODS SELECT NLEVELS ; 
TITLE1 BOLD "Number of Distinct Variable Levels (Data Cardinality)" ; 
TITLE2 BOLD "Variable Names Displayed in Alphabetical Order" ; 
PROC FREQ DATA=SASHELP.Cars NLEVELS ; 
  TABLES &mAlphabeticalVariable_List ; 
RUN ; 
TITLE1 BOLD "Origin Frequency Distribution" ; 
PROC FREQ DATA=SASHELP.Cars ; 
  TABLES Origin ; 
RUN ; 
 
ODS REGION ;  /* Row 1 Column 2 */ 
TITLE1 BOLD "Cars by Origin" ; 
PROC REPORT DATA=SASHELP.Cars(KEEP=Type Make Model Origin MSRP) ; 
  WHERE MSRP < 30000 AND Type IN ("SUV","Sports") ; 
  COLUMNS Type Origin Make Model MSRP ; 
  DEFINE Type   / ORDER          ; 
  DEFINE Origin / ORDER          ; 
  DEFINE Make   / ORDER   CENTER ; 
  DEFINE Model  / DISPLAY        ; 
  DEFINE MSRP   / DISPLAY        ; 
RUN ; 
 
ODS LAYOUT END ; 
 
ODS HTML5 CLOSE ; 

Example #3 – (2x2) Dashboard Layout with Default Settings 
PROC FREQ, PROC SGPLOT, PROC MEANS, and PROC UNIVARIATE 

 

 
 

.   .   .   .   .   .   .   .   .   .   .   . 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The four procedures that are 
used to create the dashboard are: PROC FREQ, PROC SGPLOT, PROC MEANS, and PROC UNIVARIATE. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

3. An ODS LAYOUT GRIDDED ROWS=2 COLUMNS=2 statement tells SAS to create a gridded layout consisting of one row 
and two columns. 
 

4. Multiple ODS REGION statements to tell SAS to produce the row and column of results. 
 

5. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

6. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 

Base-SAS Code: 
ODS HTML5 PATH="/home/kirklafler/Results" 
          FILE="Dashboard #2 – (2x2) Layout.html" 
          (URL=NONE) ; 
 
title1 font=impact bold h=12 c=blue "Analytics Dashboard" ; 
ods layout start rows=2 columns=2 ; 
 
ods region ;  /* Row 1 Column 1 */ 
title1 "Region Frequency Distribution" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 1 Column 2 */ 
title1 "Type BarChart" ; 
proc sgplot data=sashelp.cars ; 
  vbar Type / group=Type datalabel ; 
run ; 
 
ods region ;  /* Row 2 Column 1 */ 
title1 "Type Descriptive Statistics" ; 
proc means data=sashelp.cars n nmiss min max range mean median mode std var ; 
  class Type ; 
run ; 
 
ods region ;  /* Row 2 Column 2 */ 
title1 "Type Univariate Statistics" ; 
proc univariate data=sashelp.cars plots ; 
  class Type ; 
run ; 
title ; 
 
ods layout end ; 
ods html5 close ; 

Example #4 – (3x3) Dashboard Layout with Default Settings 
PROC CONTENTS, PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC UNIVARIATE 

 

 
 

.   .   .   .   .   .   .   .   .   .   .   . 
 

 
 
 

.   .   .   .   .   .   .   .   .   .   .   . 

Dashboard #3, continued 
 

 
 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The procedures that are used 
to create the dashboard are: PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC 
UNIVARIATE. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

1. An ODS LAYOUT GRIDDED ROWS=3 COLUMNS=3 statement tells SAS to create a gridded layout consisting of one row 
and two columns. 
 

2. Multiple ODS REGION statements to tell SAS to produce the row and column of results. 
 

3. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

4. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 

Base-SAS Code: 
ODS HTML5 PATH="/home/kirklafler/Results" 
          body="Dashboard #3 - (3x3) Layout.html" 
          (url=none) ; 
 
title1 font=impact bold h=12 c=blue "Analytics Dashboard" ; 
ods layout start rows=3 columns=3 ; 
 
ods region ;  /* Row 1 Column 1 */ 
title1 "Cars Metadata Contents" ; 
proc contents data=sashelp.cars nods ; 
run ; 
 
ods region ;  /* Row 1 Column 2 */ 
ods select nlevels ; 
title1 "NLEVELS (Data Cardinality) Results" ; 
proc freq data=sashelp.cars NLEVELS ; 

run ; 
 
ods region ;  /* Row 1 Column 3 */ 
title1 "Origin PieChart" ; 
proc sgpie data=sashelp.cars ; 
  pie Origin / datalabeldisplay=all ; 
run ; 
title1 "Type PieChart" ; 
proc sgpie data=sashelp.cars ; 
  pie Type / datalabeldisplay=all ; 
run ; 
title ; 
 
ods region ;  /* Row 2 Column 1 */ 
title1 "Origin and Type Frequency Distributions" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 2 Column 2 */ 
title1 "Origin Vertical BarChart" ; 
proc sgplot data=sashelp.cars ; 
  vbar Origin / group=Origin datalabel ; 
run ; 
title1 "Type Vertical BarChart" ; 
proc sgplot data=sashelp.cars ; 
  vbar Type   / group=Type datalabel ; 
run ; 
 
ods region ;  /* Row 2 Column 3 */ 
title1 "Origin and Type Descriptive Statistics" ; 
proc means data=sashelp.cars n nmiss min max range mean median mode std var ; 
  class Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 1 */ 
title1 "Origin and Type Frequency Distribution" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 2 */ 
title1 "PROC REPORT Results" ; 
proc report data=sashelp.cars ; 
  columns Origin Type Make Model MSRP Invoice ; 
  define Origin  / order ; 
  define Type    / order ; 
  define Make    / order ; 
  define Model   / display ; 
  define MSRP    / display format=dollar10. ; 
  define Invoice / display format=dollar10. ; 
run ; 
 
ods region ;  /* Row 3 Column 3 */ 
ods select moments ; 
title1 "Cars MOMENTS Univariate Statistics" ; 
proc univariate data=sashelp.cars ; 
  class Origin ; 
run ; 
 
ods layout end ; 
ods html5 close ; 

Example #5 – (3x3) Dashboard Layout with Custom Colors 
PROC CONTENTS, PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC UNIVARIATE 

Dashboard #4, continued 
 

 
 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The procedures that are used 
to create the dashboard are: PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC 
UNIVARIATE. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

3. An ODS LAYOUT GRIDDED ROWS=3 COLUMNS=3 statement tells SAS to create a gridded layout consisting of one row 
and two columns. 
 

4. Multiple ODS REGION statements to tell SAS to produce the row and column of results. 
 

5. When producing graphics (e.g., bar charts, pie charts, etc.) the statement styleattrs DATACOLORS=(red blue yellow 
green purple orange goldenrod cyan) tells SAS to display the bars and/or pie slices using the specified colors. 
 

6. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

7. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 

Base-SAS Code: 
ODS HTML5 PATH="/home/kirklafler/Dashboards/Results" 
          body="Dashboard #4 - (3x3) Layout with Custom Colors.html" 
          (url=none) ; 
 
title1 font=impact bold h=12 c=blue "Analytics Dashboard" ; 
ods layout start rows=3 columns=3 ; 
 
ods region ;  /* Row 1 Column 1 */ 
title1 "Cars Metadata Contents" ; 
proc contents data=sashelp.cars nods ; 
run ; 
ods region ;  /* Row 1 Column 2 */ 
ods select nlevels ; 

title1 "NLEVELS (Data Cardinality) Results" ; 
proc freq data=sashelp.cars NLEVELS ; 
run ; 
 
ods region ;  /* Row 1 Column 3 */ 
title1 "Origin Pie Chart" ; 
proc sgpie data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  pie Origin / datalabeldisplay=all ; 
run ; 
title1 "Type Pie Chart" ; 
proc sgpie data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  pie Type / datalabeldisplay=all ; 
run ; 
title ; 
 
ods region ;  /* Row 2 Column 1 */ 
title1 "Origin and Type Frequency Distributions" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 2 Column 2 */ 
title1 "Origin Vertical Bar Chart" ; 
proc sgplot data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  vbar Origin / group=Origin datalabel nooutline ; 
run ; 
title1 "Type Vertical Bar Chart" ; 
proc sgplot data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  vbar Type   / group=Type datalabel nooutline ; 
run ; 
 
ods region ;  /* Row 2 Column 3 */ 
title1 "Origin and Type Descriptive Statistics" ; 
proc means data=sashelp.cars n nmiss min max range mean median mode std var ; 
  class Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 1 */ 
title1 "Origin and Type Frequency Distribution" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 2 */ 
title1 "PROC REPORT Results" ; 
proc report data=sashelp.cars ; 
  columns Origin Type Make Model MSRP Invoice ; 
  define Origin  / order ; 
  define Type    / order ; 
  define Make    / order ; 
  define Model   / display ; 
  define MSRP    / display format=dollar10. ; 
  define Invoice / display format=dollar10. ; 
run ; 
 
ods region ;  /* Row 3 Column 3 */ 
ods select moments ; 
title1 "Cars MOMENTS Univariate Statistics" ; 
proc univariate data=sashelp.cars ; 
  class Origin ; 
run ; 
 
ods layout end ; 
ods html5 close ; 

Example #6 – (3x3) Dashboard Layout with Custom Colors and 

Enlarged Fonts 
PROC CONTENTS, PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC UNIVARIATE 

Dashboard #5, continued 
 

 
 

Key Points about Code 
 

1. SAS software provides users with numerous procedures for creating dashboard output. The procedures that are used 
to create the dashboard are: PROC FREQ, PROC SGPIE, PROC SGPLOT, PROC MEANS, PROC REPORT, and PROC 
UNIVARIATE. 
 

2. An ODS HTML5 PATH= FILE=  statement tells SAS the destination (or type of medium) to use in creating the 
dashboard including the destination path (or folder) and the name of the dashboard file. 
 

3. An ODS LAYOUT GRIDDED ROWS=3 COLUMNS=3 statement tells SAS to create a gridded layout consisting of one row 
and two columns. 
 

4. Multiple ODS REGION statements to tell SAS to produce the row and column of results. 
 

5. When producing enlarged titles and fonts (e.g., titles, footnotes, charts, etc.) the datalabeldisplay=all and 
datalabelattrs= options tell SAS to display the text associated with bars and/or pie slices using a larger size font. 
 

6. An ODS LAYOUT END statement tells SAS to terminate the dashboard layout. 
 

7. An ODS HTML5 CLOSE statement tells SAS to render the dashboard content to the dashboard file. 
 

Base-SAS Code: 
ODS HTML5 PATH="/home/kirklafler/Dashboards/Results" 
       body="Dashboard #5 - (3x3) Layout with Custom Colors and Enlarged Fonts.html" 
          (url=none) ; 
 
title1 font=impact bold h=12 c=blue "Analytics Dashboard" ; 
ods layout start rows=3 columns=3 ; 
 
ods region ;  /* Row 1 Column 1 */ 
title1 "Cars Metadata Contents" ; 
proc contents data=sashelp.cars nods ; 
run ; 
ods region ;  /* Row 1 Column 2 */ 

ods select nlevels ; 
title1 "NLEVELS (Data Cardinality) Results" ; 
proc freq data=sashelp.cars NLEVELS ; 
run ; 
 
ods region ;  /* Row 1 Column 3 */ 
title1 bold height=14pt "Origin Pie Chart" ; 
proc sgpie data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  pie Origin / datalabeldisplay=all 
               datalabelattrs=(Family="Arial" Size=12 Weight=Bold)  ; 
run ; 
title1 bold height=14pt "Type Pie Chart" ; 
proc sgpie data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  pie Type / datalabeldisplay=all 
             datalabelattrs=(Family="Arial" Size=12 Weight=Bold)  ; 
run ; 
title ; 
 
ods region ;  /* Row 2 Column 1 */ 
title1 "Origin and Type Frequency Distributions" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 2 Column 2 */ 
ods graphics on / reset=all border=off ; 
title1 bold height=16pt "Origin Vertical Bar Chart" ; 
proc sgplot data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  vbar Origin / group=Origin datalabel nooutline 
                datalabelattrs=(Family="Arial" Size=14 Weight=Bold) ; 
  xaxis fitpolicy=rotatealways labelattrs=(family='Arial Black') ; 
  xaxis valueattrs=(size=14) labelattrs=(size=14 weight=bold) ; 
  yaxis valueattrs=(size=14) labelattrs=(size=14 weight=bold) ; 
  keylegend 'bar' 'vline' / title='Origin Legend' 
                       titleattrs=(color=blue size=14pt) 
                       valueattrs=(size=14pt) noborder ; 
run ; 
title1 bold height=16pt "Type Vertical Bar Chart" ; 
proc sgplot data=sashelp.cars ; 
  styleattrs DATACOLORS=(red blue yellow green purple orange goldenrod cyan) ; 
  vbar Type   / group=Type datalabel nooutline 
                datalabelattrs=(Family="Arial" Size=12 Weight=Bold)  ; 
  xaxis fitpolicy=rotatealways labelattrs=(family='Arial Black') ; 
  xaxis valueattrs=(size=14) labelattrs=(size=14 weight=bold) ; 
  yaxis valueattrs=(size=14) labelattrs=(size=14 weight=bold) ; 
  keylegend 'bar' 'vline' / title='Type Legend' 
                       titleattrs=(color=blue size=14pt) 
                       valueattrs=(size=12pt) noborder ; 
run ; 
 
ods region ;  /* Row 2 Column 3 */ 
title1 "Origin and Type Descriptive Statistics" ; 
proc means data=sashelp.cars n nmiss min max range mean median mode std var ; 
  class Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 1 */ 
title1 "Origin and Type Frequency Distribution" ; 
proc freq data=sashelp.cars ; 
  tables Origin Type ; 
run ; 
 
ods region ;  /* Row 3 Column 2 */ 
title1 "PROC REPORT Results" ; 
proc report data=sashelp.cars ; 
  columns Origin Type Make Model MSRP Invoice ; 
  define Origin  / order ; 
  define Type    / order ; 
  define Make    / order ; 
  define Model   / display ; 

  define MSRP    / display format=dollar10. ; 
  define Invoice / display format=dollar10. ; 
run ; 
 
ods region ;  /* Row 3 Column 3 */ 
ods select moments ; 
title1 "Cars MOMENTS Univariate Statistics" ; 
proc univariate data=sashelp.cars ; 
  class Origin ; 
run ; 
 
ods graphics reset ; 
ods layout end ; 
ods html5 close ; 

 
 
 

Example #7 – Excel Multi Autofilter Dashboard Report 
PROC SORT and PROC REPORT 

Key Points about Code 
 

1. SAS Output Delivery System (ODS) provides users with the ability to create Excel dashboards, reports, and 
spreadsheet results using the ODS Excel destination. Any procedure output, such as PROC REPORT, PROC FREQ, PROC 
MEANS, PROC SGPLOT, and countless others, can be automatically written to an open Excel spreadsheet. 
 

2. The SASHELP.CARS dataset is sorted using PROC SORT in ascending order by the ORIGIN and MSRP variables. 
 

3. An ODS EXCEL FILE= statement tells SAS the path / folder where the spreadsheet is to be written along with the 
assignment of its physical name. 
 

4. A few ODS options are specified to tell SAS to create and name multiple sheets with the sheet_interval=”bygroup” 
option, assign the Origin variable’s value to each sheet with the sheet_label=”origin” option, embed titles into the 
spreadsheet with the embedded_titles="yes" option, freeze six (6) rows at the top of the spreadsheet with the 
frozen_headers=”6” option so these rows remain fixed in-place during vertical scrolling, and assign automatic filtering 
(or subsetting) to the second and third variables (or columns) with the autofilter="2-3" option. 
 

5. Produce detailed results using PROC REPORT and TITLE statements. 
 

6. An ODS Excel CLOSE statement tells SAS to render the PROC REPORT results representing the dashboard contents to 
the Excel spreadsheet file. 

 

Base-SAS Code: 
PROC SORT DATA=SASHELP.CARS 
           OUT=WORK.Cars_Sorted ; 
  BY ORIGIN MSRP ; 
RUN ; 
 
ODS Excel FILE="/Dashboards/Results/Dashboard #6 - Excel Autofilter Report.xlsx" 
          OPTIONS(sheet_interval="bygroup" 
                     sheet_label="origin" 
                 embedded_titles="yes" 
                  frozen_headers="6" 
                      autofilter="2-3") ; 
 
TITLE1 BOLD HEIGHT=12 "Excel Multi Autofilter Report" ; 
TITLE2 BOLD HEIGHT=11 "Automobiles by Origin" ; 
PROC REPORT DATA=WORK.Cars_Sorted(KEEP=Origin Type Make Model MSRP Invoice) ; 
  BY Origin ; 
  COLUMNS Origin Type Make Model MSRP Invoice ; 
  DEFINE Origin  / ORDER   "Origin of Car" ; 
  DEFINE Type    / ORDER   "Type of Car"   ; 
  DEFINE Make    / ORDER   "Make of Car"   ; 
  DEFINE Model   / DISPLAY "Car Model"     ; 
  DEFINE MSRP    / DISPLAY "MSRP"          ; 
  DEFINE Invoice / DISPLAY "Invoice Price" ; 
RUN ; 
TITLE ; 
 
ODS Excel CLOSE ; 

Example #8 – Excel Multi Autofilter Dashboard Report with           

Traffic Lighting 
PROC FORMAT, PROC SORT, and PROC REPORT 

Key Points about Code 
 

1. PROC FORMAT provides users with the ability to create and assign user-defined formats for the application of data 
standardization, color assignment, and many other valuable coding techniques. In this example, the assignment of 
colors (i.e., “Green”, “Blue”, “Orange”, and “Red”) are applied to the background in the Excel spreadsheet. 
 

2. The SASHELP.CARS dataset is sorted using PROC SORT in ascending order by the ORIGIN and MSRP variables. 
 

3. An ODS EXCEL FILE= statement tells SAS the path / folder where the spreadsheet is to be written along with the 
assignment of its physical name. 
 

4. A few ODS options are specified to tell SAS to create and name multiple sheets with the sheet_interval=”bygroup” 
option, assign the Origin variable’s value to each sheet with the sheet_label=”origin” option, embed titles into the 
spreadsheet with the embedded_titles="yes" option, freeze six (6) rows at the top of the spreadsheet with the 
frozen_headers=”6” option so these rows remain fixed in-place during vertical scrolling, and assign automatic filtering 
(or subsetting) to the second and third variables (or columns) with the autofilter="2-3" option. 
 

5. Produce detailed results using PROC REPORT and TITLE statements. 
 

6. Define MSRP as an “ANALYSIS” variable so it can be used in a COMPUTE block, along with the assignment of the 
background colors based on the MSRP value. 
 

7. An ODS Excel CLOSE statement tells SAS to render the PROC REPORT results representing the dashboard contents to 
the Excel spreadsheet file. 

 

Base-SAS Code: 
PROC FORMAT ; 
  Value MSRPFmt LOW - < 20000 = 'Green' 
              20000 - < 30000 = 'Blue' 
              30000 - < 40000 = 'Orange' 
              40000 - HIGH    = 'Red' ; 
RUN ; 
 
PROC SORT DATA=SASHELP.CARS 
           OUT=WORK.Cars_Sorted ; 
  BY Origin MSRP ; 



Dashboards Made Easy Using SAS® Software, continued 

 
Page 24 

 

RUN ; 
 
ODS Excel FILE="/Dashboards/Results/Dashboard #7 - Excel Autofilter Report with Traffic 
Lighting.xlsx" 
          OPTIONS(sheet_interval="bygroup" 
                     sheet_label="origin" 
                 embedded_titles="yes" 
                  frozen_headers="6" 
                      autofilter="2-3") ; 
 
TITLE1 BOLD HEIGHT=12 "Excel Multi Autofilter Report" ; 
TITLE2 BOLD HEIGHT=11 "Automobiles by Origin with Traffic Lighting" ; 
PROC REPORT DATA=WORK.Cars_Sorted(KEEP=Origin Type Make Model MSRP Invoice) 
   STYLE(Header)={BackGround=Blue ForeGround=White Font=(Arial, 10pt, Bold)} ; 
  BY Origin ; 
  COLUMNS Origin Make Type Model MSRP Invoice ; 
  DEFINE Origin  / ORDER    "Origin of Car" ; 
  DEFINE Type    / ORDER    "Type of Car"   ; 
  DEFINE Make    / ORDER    "Make of Car"   ; 
  DEFINE Model   / DISPLAY  "Car Model"     ; 
  DEFINE MSRP    / ANALYSIS "Vehicle MSRP" 
                   STYLE(Column)=[FontWeight=bold BackGround=MSRPFmt.] ; 
  DEFINE Invoice / DISPLAY  "Invoice Price" ; 
  COMPUTE MSRP ; 
    CALL DEFINE (_COL_,"STYLE","STYLE={ForeGround=White}") ; 
  ENDCOMP ; 
RUN ; 
 
ODS Excel close ; 

 
 

Example #9 – Traffic Lighting to Rows (Background) 
PROC SORT, ODS EXCEL, and PROC REPORT COMPUTE Block 

Key Points about Code 
 

1. PROC SORT to order the SASHELP.CARS dataset in ascending order by the ORIGIN, MAKE, TYPE, MODEL, and MSRP 
variables. 
 

2. An ODS EXCEL FILE= statement tells SAS the path / folder where the spreadsheet is to be written along with the 
assignment of its physical name, and a style definition, STYLES.MINIMAL, with the STYLE= parameter. 
 

3. Produce detailed results using PROC REPORT and TITLE statements. 
 

4. Define MSRP so it can be used in a COMPUTE block, along with the assignment of the background colors based on the 
MSRP value used in the COMPUTE block logic. In this example, the assignment of colors (i.e., “Green”, “Blue”, 
“Orange”, and “Red”) are applied to the background in the Excel spreadsheet. 
 

5. An ODS Excel CLOSE statement tells SAS to render the PROC REPORT results representing the dashboard contents to 
the Excel spreadsheet file. 

 

Base-SAS Code: 
PROC SORT DATA=SASHELP.CARS 
           OUT=WORK.CARS_SORTED ; 
  BY Origin Make Type Model MSRP ; 
RUN ; 
 
ODS Excel FILE = 'c:\Custom Row Traffic Lighting.xlsx' 
         STYLE = styles.minimal ; 
 
TITLE "Detailed Vehicle Listing" ; 
PROC REPORT DATA=WORK.Cars_Sorted ; 
  COLUMNS Origin Make Type Model MSRP ; 
  DEFINE Origin / ORDER    'Country of Origin'  ; 
  DEFINE Make   / ORDER    'Make of Vehicle'    ; 
  DEFINE Type   / ORDER    'Vehicle Type'       ; 
  DEFINE Model  / DISPLAY  'Vehicle Model'      ; 
  DEFINE MSRP   / ORDER    'Vehicle MSRP'       ; 
  DEFINE MSRP   / ORDER                         ; 
  COMPUTE MSRP ; 
    IF MSRP < 20000 THEN 
      CALL DEFINE (_ROW_,'STYLE','STYLE=[BACKGROUND=GREEN FOREGROUND=WHITE FONT_WEIGHT=BOLD]') ; 
    ELSE IF MSRP IN (20000:29999) THEN 
      CALL DEFINE (_ROW_,'STYLE','STYLE=[BACKGROUND=BLUE FOREGROUND=WHITE FONT_WEIGHT=BOLD]') ; 
    ELSE IF MSRP IN (30000:39999) THEN 
      CALL DEFINE (_ROW_,'STYLE','STYLE=[BACKGROUND=YELLOW FOREGROUND=BLACK FONT_WEIGHT=BOLD]') ; 
    ELSE IF MSRP >= 40000 THEN 
      CALL DEFINE (_ROW_,'STYLE','STYLE=[BACKGROUND=RED FOREGROUND=WHITE FONT_WEIGHT=BOLD]') ; 
  ENDCOMP ; 
RUN ; 
 
ODS Excel close ; 

Example #10 – Traffic Lighting to Column (Foreground Text) 
PROC FORMAT, ODS EXCEL, and PROC REPORT 

Key Points about Code 
 

1. PROC SORT to order the SASHELP.CARS dataset in ascending order by the ORIGIN and MSRP variables. 
 

2. PROC FORMAT to assign “custom” colors to a user-defined format. 
 

3. An ODS EXCEL FILE= statement tells SAS the path / folder where the spreadsheet is to be written along with the 
assignment of its physical name.  
 

4. Produce detailed results using PROC REPORT and TITLE statements. A style definition for the HEADER component of 
PROC REPORT is specified (Background, Foreground, and Font) with the STYLE= parameter. 
 

5. A DEFINE statement as an ANALYSIS variable with the user-defined format name, MSRPFmt., to assign the foreground 
colors based on the MSRP value specified in the PROC FORMAT. In this example, the assignment of colors (i.e., 
“Green”, “Blue”, “Orange”, and “Red”) are applied to the foreground column in the Excel spreadsheet. 
 

6. An ODS Excel CLOSE statement tells SAS to render the PROC REPORT results representing the dashboard contents to 
the Excel spreadsheet file. 

 

Base-SAS Code: 
PROC SORT DATA=SASHELP.CARS 
           OUT=WORK.CARS_SORTED ; 
  BY Origin MSRP ; 
RUN ; 
 
PROC FORMAT ; 
  Value MSRPFmt LOW - < 20000 = 'Green' 
              20000 - < 35000 = 'Blue' 
              35000 - < 50000 = 'Orange' 
              50000 - HIGH    = 'Red' ; 
RUN ; 
 
ODS Excel file=':\Column Traffic Lighting Foreground.xlsx' 
         style=styles.minimal ; 
 
PROC REPORT DATA=WORK.CARS_SORTED 
            STYLE(Header)={BackGround=Blue ForeGround=White 
                                 Font=(Arial, 10pt, Bold)} ; 
 WHERE UPCASE(Type) IN ("TRUCK","WAGON") ; 
 COLUMNS Origin Make Type Model MSRP ; 
 DEFINE MSRP / ANALYSIS 'Vehicle MSRP'   
       STYLE(Column)=[FontWeight=bold ForeGround=MSRPFmt.] ; 
RUN ; 
 
ODS Excel close ; 

Example #11 – Traffic Lighting to Column (Background) 
PROC FORMAT, ODS EXCEL, and PROC REPORT 

Key Points about Code 
 

1. PROC SORT to order the SASHELP.CARS dataset in ascending order by the ORIGIN and MSRP variables. 
 

2. PROC FORMAT to assign “custom” colors to a user-defined format. 
 

3. An ODS EXCEL FILE= statement tells SAS the path / folder where the spreadsheet is to be written along with the 
assignment of its physical name.  
 

4. Produce detailed results using PROC REPORT and TITLE statements. A style definition for the HEADER component of 
PROC REPORT is specified (Background, Foreground, and Font) with the STYLE= parameter. 
 

5. A DEFINE statement as an ANALYSIS variable with the user-defined format name, MSRPFmt., to assign the foreground 
colors based on the MSRP value specified in the PROC FORMAT. In this example, the assignment of colors (i.e., 
“Green”, “Blue”, “Orange”, and “Red”) are applied to the foreground column in the Excel spreadsheet. 
 

6. A COMPUTE block to assign the ForeGround=White color to the data in the column. 
 

7. An ODS Excel CLOSE statement tells SAS to render the PROC REPORT results representing the dashboard contents to 
the Excel spreadsheet file. 

 

Base-SAS Code: 
PROC SORT DATA=SASHELP.CARS 
           OUT=WORK.CARS_SORTED ; 
  BY Origin MSRP ; 
RUN ; 
 
PROC FORMAT ; 
  Value MSRPFmt LOW - < 20000 = 'Green' 
              20000 - < 35000 = 'Blue' 
              35000 - < 50000 = 'Orange' 
              50000 - HIGH    = 'Red' ; 
RUN ; 
 
ODS Excel file='c:\Column Traffic Lighting Background.xlsx' 
         style=styles.minimal ; 
 
PROC REPORT DATA=WORK.CARS_SORTED 
            STYLE(Header)={BackGround=Blue ForeGround=White 
                                 Font=(Arial, 10pt, Bold)} ; 
 WHERE UPCASE(Type) IN ("TRUCK","WAGON") ; 
 COLUMNS Origin Make Type Model MSRP ; 
 DEFINE MSRP / ANALYSIS 'Vehicle MSRP'  
       STYLE(Column)=[FontWeight=bold BackGround=MSRPFmt.] ; 
 COMPUTE MSRP ; 
   CALL DEFINE (_COL_,"STYLE","STYLE={ForeGround=White}") ; 
 ENDCOMP ; 
RUN ; 
 
ODS Excel close ; 

Example #12 – Listing of SAS-supplied Style Templates 
PROC TEMPLATE with LIST STYLES Statement 

 

Base-SAS Code: 
proc template ; 
  list styles ; 
run ; 

 

Results: 

Example #13 – Styles.SasWeb Style Definition 
PROC TEMPLATE with SOURCE STYLES.SasWeb Statement 

 

Base-SAS Code: 
proc template ; 
  source styles.SasWeb ; 
run ; 

 

Log Results: 
 
77         proc template ; 
78           source styles.SasWeb ; 
 define style Styles.SasWeb; 
    style fonts / 
       'TitleFont2' = ("<sans-serif>, Helvetica, sans-serif",2,bold italic) 
       'TitleFont' = ("<sans-serif>, Helvetica, sans-serif",4,bold) 
       'StrongFont' = ("<sans-serif>, Helvetica, sans-serif",2,bold) 
       'EmphasisFont' = ("<sans-serif>, Helvetica, sans-serif",2,italic) 
       'FixedEmphasisFont' = ("<monospace>, Courier, monospace",2,italic) 
       'FixedStrongFont' = ("<monospace>, Courier, monospace",2,bold) 
       'FixedHeadingFont' = ("<monospace>, Courier, monospace",2) 
       'BatchFixedFont' = ("SAS Monospace, <monospace>, Courier, monospace",2) 

       'FixedFont' = ("<monospace>, Courier, monospace",2) 
       'headingEmphasisFont' = ("<sans-serif>, Helvetica, sans-serif",2,bold italic) 
       'headingFont' = ("<sans-serif>, Helvetica, sans-serif",2,bold) 
       'docFont' = ("<sans-serif>, Helvetica, sans-serif",2); 
    class GraphFonts / 
       'GraphDataFont' = ("<sans-serif>, <MTsans-serif>",7pt) 
       'GraphUnicodeFont' = ("<MTsans-serif-unicode>",9pt) 
       'GraphValueFont' = ("<sans-serif>, <MTsans-serif>",9pt) 
       'GraphLabel2Font' = ("<sans-serif>, <MTsans-serif>",10pt) 
       'GraphLabelFont' = ("<sans-serif>, <MTsans-serif>",10pt,bold) 
       'GraphFootnoteFont' = ("<sans-serif>, <MTsans-serif>",10pt,bold) 
       'GraphTitleFont' = ("<sans-serif>, <MTsans-serif>",11pt,bold) 
       'GraphTitle1Font' = ("<sans-serif>, <MTsans-serif>",14pt,bold) 
       'GraphAnnoFont' = ("<sans-serif>, <MTsans-serif>",10pt); 
    style color_list 
       "Colors used in the default style" / 
       'fgD1' = cx666666    /* Gray            */ 
       'fgC1' = cxCCCCCC    /* Light Gray      */ 
       'fgB1' = cx000000    /* Black           */ 
       'bgA1' = cx6495ED    /* CornFlower Blue */ 
       'fgA'  = cx003399    /* Blue            */ 
       'bgA'  = cxffffff;   /* White           */ 
    style colors 
       "Abstract colors used in the default style" / 
       'headerfgemph' = color_list('bgA') 
       'headerbgemph' = color_list('bgA1') 
       'headerfgstrong' = color_list('bgA') 
       'headerbgstrong' = color_list('bgA1') 
       'headerfg' = color_list('bgA') 
       'headerbg' = color_list('bgA1') 
       'datafgemph' = color_list('fgB1') 
       'databgemph' = color_list('bgA') 
       'datafgstrong' = color_list('fgB1') 
       'databgstrong' = color_list('bgA') 
       'datafg' = color_list('fgB1') 
       'databg' = color_list('bgA') 
       'batchfg' = color_list('fgA') 
       'batchbg' = color_list('bgA') 
       'tableborder' = color_list('fgD1') 
       'tablebg' = cxcccccc 
       'notefg' = color_list('fgA') 
       'notebg' = color_list('bgA') 
       'bylinefg' = color_list('fgA') 
       'bylinebg' = color_list('bgA') 
       'captionfg' = color_list('fgA') 
       'captionbg' = color_list('bgA') 
       'proctitlefg' = color_list('fgA') 
       'proctitlebg' = color_list('bgA') 
       'titlefg' = color_list('fgA') 
       'titlebg' = color_list('bgA') 
       'systitlefg' = color_list('fgA') 
       'systitlebg' = color_list('bgA') 
       'contentfg' = color_list('fgA') 
       'contentbg' = color_list('bgA') 
       'docfg' = color_list('fgA') 
       'docbg' = color_list('bgA'); 
 
 .  .  .  .  .  .  .  .  . 
 
end; 
 NOTE: Path 'Styles.SasWeb' is in: SASHELP.TMPL_EN (via SASHELP.TMPLMST). 
 79         run ; 

Example #14 – Single Column Black & White Dashboard 
PROC FORMAT, PROC SORT, and PROC REPORT 

Key Points about Code 
 

1. PROC TEMPLATE provides users with the ability to create and/or customize the appearance of tabular SAS output. A 
new styles.SasWeb_White_Black template using PROC TEMPLATE is created by modifying two parameters (‘fgB1’ and 
‘bgA’) in the style color_list section. 
 

2. An ODS HTML5 FILE= statement tells SAS the path / folder where the output is to be written along with the 
assignment of its physical name. 
 

3. A TITLE statement is specified to display the name of the dashboard. 
 

4. An ODS LAYOUT statement is specified to tell SAS to define a 1 row x 1 column layout. 
 

5. An ODS REGION statement is specified to indicate the beginning of output results. 
 

6. A PROC FREQ, three PROC SGPLOTs, and a PROC MEANS is specified. 
 



Dashboards Made Easy Using SAS® Software, continued 

 
Page 36 

 

7. An ODS LAYOUT CLOSE statement is specified to terminate the layout of output results. 
 

8. An ODS HTML5 CLOSE statement tells SAS to render the output results representing the dashboard contents to the 
HTML5 file. 

 

Base-SAS Code: 
proc template ; 
  define style Styles.Sasweb_White_Black ; 
    style color_list 
    "Colors used in the default style" / 
      'fgD1'     = cx666666 /* Gray             */ 
      'fgC1'     = cxCCCCCC /* Light Gray       */ 
      'fgB1'     = cxFFFFFF /* White            */ 
      'bgA1'     = cx6495ED /* CornFlower Blue  */ 
      'fgA'      = cx003399 /* Dark Blue        */ 
      'bgA'      = cx000000 /* Black            */ ; 
  end ; 
run ; 
 
ods html5 style=styles.Sasweb_White_Black 
          path="/home/kirklafler/Dashboards/Results" 
          body="Dashboard - Color (White-Black).html" 
          (url=none) ; 
 
title1 font=impact bold j=c h=12 c=black "Analytics Dashboard" ; 
 
ODS LAYOUT GRIDDED ROWS=1 COLUMNS=1 ; /* Design HTML 1x1 Layout */ 
 
options center ; /* Center the Results */ 
ods region ;  /* Start of Output Results */ 
title1 "SASHELP.CARS Frequency Distribution for Origin and Type" ; 
proc freq data=SASHELP.CARS NLEVELS ; 
  table Origin Type ; 
run ; 
 
title1 "Origin BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  vbar Origin / group=Origin datalabel ; 
run ; 
 
title1 "Type BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  vbar Type / group=Type datalabel ; 
run ; 
 
title1 "Origin by Type Cluster BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  vbar Origin / group=Type response=MSRP stat=mean groupdisplay=cluster datalabel ; 
run ; 
 
title1 "Descriptive Statistics for MSRP and Invoice by Origin" ; 
footnote1 j=l "Layout: HTML-fgB1-CXFFFFFF-bgA-CX000000 (White/Black)" ; 
proc means data=SASHELP.CARS n nmiss min max range mean median mode std var ; 
  class Origin Type ; 
run ; 
title ; 
 
ods layout end ; /* Terminate the Layout of Output Results */ 
ods html5 close ; 

Example #15 – Single Column Black & Burgundy Dashboard 
PROC FORMAT, PROC SORT, and PROC REPORT 

Key Points about Code 
 

1. A new style template, Styles.Sasweb_Black_Burgundy, is created with PROC TEMPLATE that inherits the attributes of 
its parent template, Styles.SASWEB. The new template replaces two parameters (‘fgB1’ and ‘bgA1’) in the style 
color_list section. 
 

2. An ODS HTML5 FILE= statement tells SAS the path / folder where the output is to be written along with the 
assignment of its physical name. 
 

3. A TITLE statement is specified to display the name of the dashboard. 
 

4. An ODS LAYOUT statement is specified to tell SAS to define a 1 row x 1 column layout. 

5. An ODS REGION statement is specified to indicate the beginning of output results. 
 

6. A PROC FREQ, three PROC SGPLOTs, and a PROC MEANS is specified. 
 

7. An ODS LAYOUT CLOSE statement is specified to terminate the layout of output results. 
 

8. An ODS HTML5 CLOSE statement tells SAS to render the output results representing the dashboard contents to the 
HTML5 file. 

 

Base-SAS Code: 
proc template ; 
  define style Styles.Sasweb_Black_Burgundy ; 
    parent = Styles.SASWEB ; 
      replace color_list /  
        'fgD1' = cx666666 /* Gray             */ 
        'fgC1' = cxCCCCCC /* Light Gray       */ 
        'fgB1' = CX000000 /* Black            */ 
        'bgA1' = CX800020 /* Burgundy         */ 
        'fgA'  = CX000000 /* Black            */ 
        'bgA'  = CXFFFFFF /* White            */ ; 
  end ; 
run ; 
 
ods html5 style=styles.Sasweb_Black_Burgundy 
          path="/home/kirklafler/Dashboards/Results" 
          body="Dashboard - Color (Black-Burgundy) with STYLEATTRS.html" 
          (url=none) ; 
 
title1 font=impact bold j=c h=12 c=Black "Analytics Dashboard" ; 
 
ODS LAYOUT GRIDDED ROWS=1 COLUMNS=1 ; /* Design HTML 1x1 Layout */ 
 
options center ; /* Center the Results */ 
ods region ;  /* Start of Output Results */ 
title1 "SASHELP.CARS Frequency Distribution for Origin and Type" ; 
proc freq data=SASHELP.CARS NLEVELS ; 
  table Origin Type ; 
run ; 
 
title1 "Origin BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  styleattrs DATACOLORS=(CX800020 Indigo Plum Magenta PaleVioletRed Crimson) ; 
  vbar Origin / group=Origin datalabel ; 
run ; 
 
title1 "Type BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  styleattrs DATACOLORS=(CX800020 Indigo Plum Magenta PaleVioletRed Crimson) ; 
  vbar Type / group=Type datalabel ; 
run ; 
 
title1 "Origin by Type Cluster BarChart" ; 
proc sgplot data=SASHELP.CARS ; 
  styleattrs DATACOLORS=(CX800020 Indigo Plum Magenta PaleVioletRed Crimson) ; 
  vbar Origin / group=Type response=MSRP stat=mean groupdisplay=cluster datalabel ; 
run ; 
 
title1 "Descriptive Statistics for MSRP and Invoice by Origin" ; 
footnote1 j=l "Layout: HTML-bgA1-CX800020-fgA-CX000000 (Black-Burgundy) with STYLEATTRS" ; 
proc means data=SASHELP.CARS n nmiss min max range mean median mode std var ; 
  class Origin Type ; 
run ; 
title ; 
 
ods layout end ; /* Terminate the Layout of Output Results */ 
ods html5 close ; 

Conclusion 

Organizations around the globe develop business intelligence and analytics dashboards to display the status of “point-in-time” 
metrics and key performance indicators. An effectively designed dashboard extracts real-time data from multiple sources for 
the purpose of highlighting important information, numbers, tables, statistics, metrics, performance scorecards and other 
essential content. This paper explored essential rules for “good” dashboard design, the metrics frequently used in dashboards, 
and the use of best practice programming techniques in the design of aesthetically pleasing dashboards using SAS® software. 
Readers were shown programming techniques to create quick and easy dashboards using Base-SAS® software including PROC 
SQL, macro, Output Delivery System (ODS), ODS HTML, ODS Excel, ODS Layout, ODS Statistical Graphics, PROC SGPLOT, and 
PROC SGPIE. 

